

神戸市立工業高等専門学校
電気工学科／電子工学科
専門科目「数値解析」

2018.1.26

演習7

山浦 剛 (tyamaura@riken.jp)

講義資料ページ

- http://climate.aics.riken.jp/members/yamaura/numerical_analysis.html

復習：ガウスの消去法

➤ $N, a_{i,j}, y_i$ を設定 $\Rightarrow Ax = y$ の問題設定

➤ $k := 1, 2, \dots, N - 1$ の順に

- (部分ピボット選択)
 $|a_{i,k}|$ ($i = k, k + 1, \dots, N$)のうち、
最大値を $|a_{l,k}|$ とする。 $l \neq k$ ならば、
係数 $a_{i,k}$ および y_k を入れ替える。

➤ $i := k + 1, k + 2, \dots, N$ の順に

- $\alpha := a_{i,k} / a_{k,k}$
- $j := k + 1, k + 2, \dots, N$ の順に
- $a_{i,j} := a_{i,j} - \alpha a_{k,j}$
- を繰り返す
- $y_i := y_i - \alpha y_k$
- を繰り返す

を繰り返す

$$x_N := y_N / a_{N,N}$$

➤ $i := N - 1, N - 2, \dots, 1$ の順に

- $x_i := (y_i - \sum_{k=i+1}^N a_{i,k} x_k) / a_{i,i}$
- を繰り返す

➤ 最終的に求めたい解 x が算出できる。

例題1：ガウスの消去法

- ガウスの消去法を用いて、以下の連立1次方程式を解く。

- $$\begin{pmatrix} 1 & -2 & 3 \\ 2 & 1 & 0 \\ 1 & 2 & -1 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} 1 \\ 5 \\ 5 \end{pmatrix}$$

- アルゴリズムをプログラム化。
 - 部分ピボット選択は含まれていない。

```
A(1,1) = 1.0d0  
A(1,2) = -2.0d0  
A(1,3) = 3.0d0  
A(2,1) = 2.0d0  
A(2,2) = 1.0d0  
A(2,3) = 0.0d0  
A(3,1) = 1.0d0  
A(3,2) = 2.0d0  
A(3,3) = -1.0d0
```

```
Y(1) = 1.0d0  
Y(2) = 5.0d0  
Y(3) = 5.0d0
```

```
do k = 1, N-1  
  do i = k+1, N  
    b = A(i,k) / A(k,k)  
    do j = k+1, N  
      A(i,j) = A(i,j) - b * A(k,j)  
    end do  
    Y(i) = Y(i) - b * Y(k)  
  end do  
end do
```

```
X(N) = Y(N) / A(N,N)  
do i = N-1, 1, -1  
  c = 0.0d0  
  do k = i+1, N  
    c = c + A(i,k) * X(k)  
  end do  
  X(i) = ( Y(i) - c ) / A(i,i)  
end do
```

復習：3重対角行列のLU分解

➤ $N, a_{i,j}, y_i$ を設定 $\Rightarrow Ax = y$ の問題設定

➤ LU分解(係数行列 A を L と U に分解)

➤
$$\begin{aligned} d_1 &:= a_1 \\ i &:= 2, 3, \dots, N \text{の順に} \\ &\quad l_i := b_i / d_{i-1} \\ &\quad d_i := a_i - l_i c_{i-1} \\ &\quad \text{を繰り返す} \end{aligned}$$

➤ 前進代入($Lz = y$ を解く)

➤
$$\begin{aligned} z_1 &:= y_1 \\ i &:= 2, 3, \dots, N \text{の順に} \\ &\quad z_i := y_i - l_i z_{i-1} \\ &\quad \text{を繰り返す} \end{aligned}$$

➤ 後退代入($Ux = z$ を解く)

➤
$$\begin{aligned} x_N &:= z_N / d_N \\ i &:= N-1, N-2, \dots, 1 \text{の順に} \\ &\quad x_i := (z_i - c_i x_{i+1}) / d_i \\ &\quad \text{を繰り返す} \end{aligned}$$

例題2：3重対角行列のLU分解

- LU分解を用いて、以下の連立1次方程式を解く。

➤
$$\begin{pmatrix} 2 & -1 & 0 \\ -1 & 2 & -1 \\ 0 & -1 & 2 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} -1 \\ 2 \\ 1 \end{pmatrix}$$

- 係数行列を次のように考える。

➤
$$\begin{pmatrix} a_1 & c_1 & & & 0 \\ b_2 & a_2 & c_2 & & \\ & \ddots & \ddots & \ddots & \\ & & b_{N-1} & a_{N-1} & c_{N-1} \\ 0 & & & b_N & a_N \end{pmatrix}$$

```
a(:) = 2.0d0  
b(:) = -1.0d0  
c(:) = -1.0d0
```

```
Y(1) = -1.0d0  
Y(2) = 2.0d0  
Y(3) = 1.0d0
```

```
d(1) = a(1)  
do i = 2, N  
  l(i) = b(i) / d(i-1)  
  d(i) = a(i) - l(i) * c(i-1)  
end do
```

```
Z(1) = Y(1)  
do i = 2, N  
  Z(i) = Y(i) - l(i) * Z(i-1)  
end do
```

```
X(N) = Z(N) / d(N)  
do i = N-1, 1, -1  
  X(i) = ( Z(i) - c(i) * X(i+1) ) / d(i)  
end do
```

課題

1. 次の連立1次方程式について、例題1を参考に、部分ピボット選択のアルゴリズムを追加して解け。

$$\begin{pmatrix} 2 & -2 & 1 \\ 1 & -1 & 2 \\ -1 & 3 & 1 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \end{pmatrix} = \begin{pmatrix} -9 \\ -3 \\ 12 \end{pmatrix}$$

2. 次の連立1次方程式について、3重対角行列のLU分解を用いて解け。

$$\begin{pmatrix} 4 & 1 & 0 & 0 & 0 \\ 2 & 4 & 1 & 0 & 0 \\ 0 & 2 & 4 & 1 & 0 \\ 0 & 0 & 2 & 4 & 1 \\ 0 & 0 & 0 & 2 & 4 \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \\ x_5 \end{pmatrix} = \begin{pmatrix} 2 \\ -1 \\ -1 \\ 0 \\ 3 \end{pmatrix}$$

3. 次の3重対角行列を含む連立1次方程式を解け。ただし $\alpha = \frac{1}{2}$, $n = 100$ とし、 $y_i = 1 - \left| \frac{2i}{n} - 1 \right|$ ($i = 1, 2, \dots, n$) とする。

$$\circ \quad \begin{bmatrix} 1+2\alpha & -\alpha & & & & \\ -\alpha & 1+2\alpha & -\alpha & & & \\ & -\alpha & 1+2\alpha & -\alpha & & \\ & & \ddots & \ddots & \ddots & \\ & & & -\alpha & 1+2\alpha & -\alpha \\ \mathbf{0} & & & & -\alpha & 1+2\alpha \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ \vdots \\ x_{n-1} \\ x_n \end{bmatrix} = \begin{bmatrix} y_1 \\ y_2 \\ y_3 \\ \vdots \\ y_{n-1} \\ y_n \end{bmatrix}$$

提出方法

- ✂切: 2018/02/02(金) 講義開始まで
- メールにプログラムを添付
 - 主題: 演習7レポート(学籍番号)
 - 宛先: tyamaura@riken.jp
 - 本文: なくてもOK
 - 添付: 学籍番号_課題番号.f90 を3ファイル
 - 課題7-1: r???????_kadai07-1.f90
 - 課題7-2: r???????_kadai07-2.f90
 - 課題7-3: r???????_kadai07-3.f90

課題1: 解答例

- サンプル7-1を書き換え
- 部分ピボット選択アルゴリズムの追加

```
program Gauss
implicit none

integer, parameter :: N = 3
```

```
real(8) :: A(N,N)
real(8) :: X(N)
real(8) :: Y(N)
```

```
real(8) :: b, c, Amax
real(8) :: Ytmp, Atmp(N)
```

```
integer :: i, j, k, l
```

```
A(1,1) = 2.0d0
A(1,2) = -2.0d0
A(1,3) = 1.0d0
A(2,1) = 1.0d0
A(2,2) = -1.0d0
A(2,3) = 2.0d0
A(3,1) = -1.0d0
A(3,2) = 3.0d0
A(3,3) = 1.0d0
```

```
Y(1) = -9.0d0
Y(2) = -3.0d0
Y(3) = 12.0d0
```

```
do k = 1, N-1
```

```
  Amax = 0.0d0
  do i = k, N
    if( abs(A(i,k)) > abs(Amax) ) then
      l = i
      Amax = A(l,k)
      Atmp(k:N) = A(l,k:N)
    end if
  end do
  if( l /= k ) then
    A(l,k:N) = A(k,k:N)
    A(k,k:N) = Atmp(k:N)
    Ytmp = Y(l)
    Y(l) = Y(k)
    Y(k) = Ytmp
  end if
```

```
    do i = k+1, N
      b = A(i,k) / A(k,k)
      do j = k+1, N
        A(i,j) = A(i,j) - b * A(k,j)
      end do
      Y(i) = Y(i) - b * Y(k)
    end do
  end do
```

```
X(N) = Y(N) / A(N,N)
do i = N-1, 1, -1
  c = 0.0d0
  do k = i+1, N
    c = c + A(i,k) * X(k)
  end do
  X(i) = ( Y(i) - c ) / A(i,i)
end do
```

```
do i = 1, N
  write(*,*) ' X(',i,') = ',X(i)
end do
```

```
stop
end program
```


課題2: 解答例

- サンプル7-2を書き換え
- 問題設定の変更
- アルゴリズムの変更はなし

```
program LU_decomp
  implicit none

  integer, parameter :: N = 5

  real(8) :: a(N)
  real(8) :: b(N)
  real(8) :: c(N)
  real(8) :: d(N)
  real(8) :: l(N)

  real(8) :: X(N)
  real(8) :: Y(N)
  real(8) :: Z(N)

  integer :: i

  a(:) = 4.0d0
  b(:) = 2.0d0
  c(:) = 1.0d0

  Y(1) = 2.0d0
  Y(2) = -1.0d0
  Y(3) = -1.0d0
  Y(4) = 0.0d0
  Y(5) = 3.0d0
```

```
  d(1) = a(1)
  do i = 2, N
    l(i) = b(i) / d(i-1)
    d(i) = a(i) - l(i) * c(i-1)
  end do

  Z(1) = Y(1)
  do i = 2, N
    Z(i) = Y(i) - l(i) * Z(i-1)
  end do

  X(N) = Z(N) / d(N)
  do i = N-1, 1, -1
    X(i) = ( Z(i) - c(i) * X(i+1) ) / d(i)
  end do

  do i = 1, N
    write(*,*) ' X(' ,i,') = ',X(i)
  end do

  stop
end program
```

課題3: 解答例

- サンプル7-2を書き換え
- 問題設定の変更
- アルゴリズムの変更はなし

```
program LU_decomp  
implicit none
```

```
integer, parameter :: N = 100
```

```
real(8) :: a(N)  
real(8) :: b(N)  
real(8) :: c(N)  
real(8) :: d(N)  
real(8) :: l(N)
```

```
real(8) :: X(N)  
real(8) :: Y(N)  
real(8) :: Z(N)
```

```
integer :: i
```

```
a(:) = 2.0d0  
b(:) = -0.5d0  
c(:) = -0.5d0
```

```
do i = 1, N  
  Y(i) = 1.0d0 - abs( 2.0d0 * dble(i) / N - 1.0d0 )  
end do
```

```
d(1) = a(1)  
do i = 2, N  
  l(i) = b(i) / d(i-1)  
  d(i) = a(i) - l(i) * c(i-1)  
end do
```

```
Z(1) = Y(1)  
do i = 2, N  
  Z(i) = Y(i) - l(i) * Z(i-1)  
end do
```

```
X(N) = Z(N) / d(N)  
do i = N-1, 1, -1  
  X(i) = ( Z(i) - c(i) * X(i+1) ) / d(i)  
end do
```

```
do i = 1, N  
  write(*,*) 'X(',i,') = ',X(i)  
end do
```

```
stop  
end program
```

課題3補足

- 時間ループを作り、YベクトルをXベクトルで置き換え。
- 拡散方程式の陰公式を解くことに相当する。

```
program LU_decomp
implicit none

integer, parameter :: N = 100

real(8) :: a(N)
real(8) :: b(N)
real(8) :: c(N)
real(8) :: d(N)
real(8) :: l(N)

real(8) :: X(N)
real(8) :: Y(N)
real(8) :: Z(N)

integer :: i, t

a(:) = 2.0d0
b(:) = -0.5d0
c(:) = -0.5d0

do i = 1, N
  Y(i) = 1.0d0 - abs( 2.0d0 * dble(i) / N - 1.0d0 )
end do

do t = 1, 2000
  write(*,*) Y(:)

  d(1) = a(1)
  do i = 2, N
    l(i) = b(i) / d(i-1)
    d(i) = a(i) - l(i) * c(i-1)
  end do

  Z(1) = Y(1)
  do i = 2, N
    Z(i) = Y(i) - l(i) * Z(i-1)
  end do

  X(N) = Z(N) / d(N)
  do i = N-1, 1, -1
    X(i) = ( Z(i) - c(i) * X(i+1) ) / d(i)
  end do

  Y(:) = X(:)
end do

stop
end program
```

