神戸市立工業高等専門学校
電気工学科／電子工学科
専門科目「数値解析」

2017.11.10

演習5

山浦 剛（tyamaura@riken.jp）

講義資料ページ

- http://climate.aics.riken.jp/members/yamaura/numerical_analysis.html

復習：オイラー法

- $\frac{1}{\Delta t}\left\{Y_{j+1} - Y_j\right\} = f\left(t_j, Y_j\right)$



復習：ホイン法（改良オイラー法）

- $\frac{1}{\Delta t}\{Y_{j+1} - Y_j\} = \frac{1}{2}(k_1 + k_2)$, $k_1 = f\left(t_j, Y_j\right)$, $k_2 = f\left(t_j + \Delta t, Y_j + \Delta t k_1\right)$



復習: ルンゲ･クッタ法

- $\frac{1}{\Delta t}\{Y_{j+1} - Y_j\} = \frac{1}{6}(k_1 + 2k_2 + 2k_3 + k_4)$
  - $k_1 = f(t_j, Y_j),\ k_2 = f\left(t_j + \frac{\Delta t}{2}, Y_j + \frac{\Delta t}{2}k_1\right),\ k_3 = f\left(t_j + \frac{\Delta t}{2}, Y_j + \frac{\Delta t}{2}k_2\right),\ k_4 = f(t_j + \Delta t, Y_j + \Delta t k_3)$



例題1

- $\frac{dy}{dt} = y(5-y),\ y(0) = 1$ の初期値問題を、差分方程式によって$t = 2$ まで解く（pp.80）。
  - サンプル1～3が対応（オイラー法、ホイン法、ルンゲ・クッタ法による数値解を出力するプログラム）
- $\Delta t = 0.5$における各差分法による数値解を右図に示す。
  - オイラー法（赤）、ホイン法（緑）、ルンゲ・クッタ法（青）、厳密解（紫）
- Δtを変えた場合にどうなるかを確かめてみよう。



例題2

- $\frac{d^2y}{dt^2} = -y, y(0) = 1, \frac{dy(0)}{dt} = 0$ の初期値問題を、差分方程式によって$t = 20$ まで解く(pp.82)。
  - サンプル4が対応(オイラー法による数値解を出力するプログラム)
- $\Delta t = 0.1$におけるオイラー法による数値解を右図に示す。
  - オイラー法(赤)、厳密解(緑)
- Δtを変えた場合にどうなるかを確かめてみよう。
- 余裕があれば、この初期値問題をホイン法やルンゲ・クッタ法で解いてみよう。



課題

- 次の微分方程式の初期値問題が与えられたとき、以下の問いに答えよ。
  - $\frac{dy}{dt} = y, \quad y(0) = 1$

1. $t = 0$から$t = 5$の間で$\Delta t = 5 \times 10^{-1}, 5 \times 10^{-2}, 5 \times 10^{-3}$の場合のオイラー法による差分解をそれぞれ図で示せ。
2. 課題5-1について、ホイン法による差分解をそれぞれ図で示せ。
3. 課題5-1について、ルンゲ･クッタ法による差分解をそれぞれ図で示せ。

- 次の微分方程式の初期値問題が与えられたとき、以下の問いに答えよ。
  - $\frac{d^2y}{dt^2} + 2k\frac{dy}{dt} + \omega^2 y = 0, \quad y(0) = 1, \quad \frac{d}{dt}y(0) = 0$

4. $k = 5, \omega = 4$のとき、$t = 0$から$t = 2$の間で$\Delta t = 0.4, 0.2, 0.1$の場合のオイラー法による差分解をそれぞれ図で示せ。
5. 課題5-4について、ホイン法による差分解をそれぞれ図で示せ。
6. 課題5-4について、ルンゲ･クッタ法による差分解をそれぞれ図で示せ。

提出方法

- 〆切：2017/11/17（金）講義開始まで
- メールにプログラムを添付
  - 主題：演習5レポート（学籍番号）
  - 宛先：tyamaura@riken.jp
  - 本文：なくてもOK
  - 添付：学籍番号_課題番号.f90 を6ファイル
    - 課題5-1：r???????_kadai05-1.f90
    - 課題5-2：r???????_kadai05-2.f90
    - 課題5-3：r???????_kadai05-3.f90
    - 課題5-4：r???????_kadai05-4.f90
    - 課題5-5：r???????_kadai05-5.f90
    - 課題5-6：r???????_kadai05-6.f90
  - 備考：パラメータΔtの値は問わない（いずれかに設定されてあれば良い）。

課題1：解答例

- サンプル5-1を書き換え

```
program euler
  implicit none

  integer, parameter :: N = 1000
  real(8), parameter :: tmax = 5.0d0

  integer :: i
  real(8) :: dt = tmax/N
  real(8) :: t, y

  t = 0.0d0
  y = 1.0d0

  write(*,'(a6,2a19)') 'step', 't', 'y'

  do i = 0, N
    write(*,'(i6,1x,2(f18.12,1x))') i, t, y

    t = t + dt
    y = y + dt * y
  end do

end program
```



課題2：解答例

- 係数k1, k2の導出が肝要

```
program heun
  implicit none

  integer, parameter :: N = 1000
  real(8), parameter :: tmax = 5.0d0

  integer :: i
  real(8) :: dt = tmax/N
  real(8) :: t, y, k1, k2

  t = 0.0d0
  y = 1.0d0
  k1 = 0.0d0
  k2 = 0.0d0

  write(*,'(a6,4a19)') 'step', 't', 'y', 'k1', 'k2'

  do i = 0, N
    write(*,'(i6,1x,4(f18.12,1x))') i, t, y, k1, k2

    t  = t + dt
    k1 = y
    k2 = y + k1 * dt
    y  = y + dt / 2.0d0 * ( k1 + k2 )
  end do

end program
```



課題3：解答例

- ホイン法の拡張で対応できる
- 係数k2, k3はdt/2に注意

```
program runge_kutta
  implicit none

  integer, parameter :: N = 1000
  real(8), parameter :: tmax = 5.0d0

  integer :: i
  real(8) :: dt = tmax/N
  real(8) :: t, y, k1, k2, k3, k4

  t = 0.0d0
  y = 1.0d0
  k1 = 0.0d0
  k2 = 0.0d0
  k3 = 0.0d0
  k4 = 0.0d0

  write(*,'(a6,6a19)') 'step', 't', 'y', 'k1', 'k2', 'k3', 'k4'

  do i = 0, N
    write(*,'(i6,1x,6(f18.12,1x))') i, t, y, k1, k2, k3, k4

    t  = t + dt
    k1 = y
    k2 = y + k1 * dt * 0.5d0
    k3 = y + k2 * dt * 0.5d0
    k4 = y + k3 * dt
    y  = y + dt / 6.0d0 * ( k1 + 2.0d0 * k2 + 2.0d0 * k3 + k4 )
  end do

end program
```



課題4：解答例

- サンプル5-4の書き換え
- 一次変数p1, p2に新しいy1, y2の値を保存しておき、最後に書き戻すようにする
- これをy1 = y1 + dt * y2 としてしまうと、その次のy2を計算するときにy1が更新されてしまっており、正しい計算ができなくなる

```
program euler
  implicit none

  integer, parameter :: N = 20
  real(8), parameter :: tmax = 2.0d0

  integer :: i
  real(8) :: dt = tmax/N
  real(8) :: t, y1, y2
  real(8) :: p1, p2

  t = 0.0d0
  y1 = 1.0d0
  y2 = 0.0d0

  write(*,'(a6,3a19)') 'step', 't', 'y1', 'y2'

  do i = 0, N
    write(*,'(i6,1x,3(f18.12,1x))') i, t, y1, y2

    t = t + dt
    p1 = y1 + dt * y2
    p2 = y2 + dt * ( -16.0d0 * y1 - 10.0d0 * y2 )
    y1 = p1
    y2 = p2
  end do

end program
```



課題5：解答例

- ホイン法の2階常微分方程式の初期値問題
- これは連立1階常微分方程式への書き換えが可能
- 係数k11, k12, k21, k22の4つ
- ホイン法ではk11, k21の平均で新しいy1を、k21, k22の平均で新しいy2を計算する
- 似たような名称の変数とする場合、取り違えに注意する

```
program heun
 implicit none

 integer, parameter :: N = 5
 real(8), parameter :: tmax = 2.0d0

 integer :: i
 real(8) :: dt = tmax/N
 real(8) :: t, y1, y2
 real(8) :: k11, k12, k21, k22

 t = 0.0d0
 k11 = 0.0d0
 k12 = 0.0d0
 k21 = 0.0d0
 k22 = 0.0d0
 y1  = 1.0d0
 y2  = 0.0d0

 write(*,'(a6,3a19)') 'step', 't', 'y1', 'y2'

 do i = 0, N
  write(*,'(i6,1x,3(f18.12,1x))') i, t, y1, y2

  t = t + dt
  k11 = y2
  k12 = -16.0d0 * y1 - 10.0d0 * y2
  k21 = y2 + dt * k12
  k22 = -16.0d0 * ( y1 + dt * k11 ) - 10.0d0 * ( y2 + dt * k12 )
  y1  = y1 + dt / 2.0d0 * ( k11 + k21 )
  y2  = y2 + dt / 2.0d0 * ( k12 + k22 )
 end do

end program
```



課題6：解答例

- 課題5の拡張、ルンゲ・クッタ法
- 係数をホイン法同様にy1, y2それぞれに対応するk1, k2, k3, k4を用意する
- それぞれの係数は次の係数を計算するときに使用されるので、計算する順番に注意する

```
program runge_kutta
 implicit none

 integer, parameter :: N = 20
 real(8), parameter :: tmax = 2.0d0

 integer :: i
 real(8) :: dt = tmax/N
 real(8) :: t, y1, y2
 real(8) :: k11, k12, k21, k22, k31, k32, k41, k42

 t = 0.0d0
 k11 = 0.0d0
 k12 = 0.0d0
 k21 = 0.0d0
 k22 = 0.0d0
 k31 = 0.0d0
 k32 = 0.0d0
 k41 = 0.0d0
 k42 = 0.0d0
 y1  = 1.0d0
 y2  = 0.0d0

 write(*,'(a6,3a19)') 'step', 't', 'y1', 'y2'

 do i = 0, N
  write(*,'(i6,1x,3(f18.12,1x))') i, t, y1, y2

  t = t + dt
  k11 = y2
  k12 = -16.0d0 * y1 - 10.0d0 * y2
  k21 = y2 + dt / 2.0d0 * k12
  k22 = -16.0d0 * ( y1 + dt / 2.0d0 * k11 ) &
         - 10.0d0 * ( y2 + dt / 2.0d0 * k12 )
  k31 = y2 + dt / 2.0d0 * k22
  k32 = -16.0d0 * ( y1 + dt / 2.0d0 * k21 ) &
         - 10.0d0 * ( y2 + dt / 2.0d0 * k22 )
  k41 = y2 + dt * k32
  k42 = -16.0d0 * ( y1 + dt * k31 ) - 10.0d0 * ( y2 + dt * k32 )
  y1  = y1 + dt / 6.0d0 * ( k11 + 2.0d0 * k21 + 2.0d0 * k31 + k41 )
  y2  = y2 + dt / 6.0d0 * ( k12 + 2.0d0 * k22 + 2.0d0 * k32 + k42 )
 end do

end program
```